<commit_before>{ lib, stdenv, fetchurl, tcl, openssl }:

stdenv.mkDerivation rec {
  pname = "tcltls";
  version = "1.6.7";

  src = fetchurl {
    url = "mirror://sourceforge/tls/tls${version}-src.tar.gz";
    sha256 = "1f53sfcnrridjl5ayrq1xrqkahs8khf8c3d0m2brndbhahzdw6ai";
  };

  buildInputs = [ tcl openssl ];

  configureFlags = [
    "--with-tcl=${tcl}/lib"
    "--with-tclinclude=${tcl}/include"
    "--with-ssl-dir=${openssl.dev}"
  ];

  preConfigure = ''
    configureFlags="--exec_prefix=$prefix $configureFlags"
  '';

  passthru = {
    libPrefix = "tls${version}";
  };

  meta = {
    homepage = "http://tls.sourceforge.net/";
    description = "An OpenSSL / RSA-bsafe Tcl extension";
    license = lib.licenses.tcltk;
    platforms = lib.platforms.unix;
  };
}<commit_msg>tcltls: Simplify package with tcl.mkTclDerivation<commit_after>{ lib, stdenv, fetchurl, tcl, openssl }:

tcl.mkTclDerivation rec {
  pname = "tcltls";
  version = "1.6.7";

  src = fetchurl {
    url = "mirror://sourceforge/tls/tls${version}-src.tar.gz";
    sha256 = "1f53sfcnrridjl5ayrq1xrqkahs8khf8c3d0m2brndbhahzdw6ai";
  };

  buildInputs = [ openssl ];

  configureFlags = [
    "--with-ssl-dir=${openssl.dev}"
  ];

  meta = {
    homepage = "http://tls.sourceforge.net/";
    description = "An OpenSSL / RSA-bsafe Tcl extension";
    license = lib.licenses.tcltk;
    platforms = lib.platforms.unix;
  };
}
{'fraction_non_alphanumeric': 0.1660727013542409, 'fraction_numerical': 0.024233784746970778, 'mean_word_length': 3.2804878048780486, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2131333', 'n_tokens_mistral': 583, 'n_tokens_neox': 554, 'n_words': 83}